Maple and Calculus Operations

Although our initial Intro to Maple included discussion of a number of Maple functions related to calculus, you may not have noticed that fact if you didn't look carefully at what Maple can do. This handout is, in the first instance, just a summary of Maple's calculus-related functionality. As always, I'm only scratching the surface. The Maple **help** command (say **?diff;** to get help on **diff**, for instance) offers a lot more information on any of the commands used in this worksheet.

Basic Commands

Limits

Maple's command for limits is, logically enough, **limit**.

```
> limit(sin(x)/x, x=0);
```

$$1$$

```
> limit(1/x, x=0);
```

$$\mathit{undefined}$$

```
> limit(1/x^2, x=0);
```

$$\infty$$

```
> limit(sin(1/x), x=0);
```

$$-1..1$$

It is useful to plot all the functions above, and to make sure you understand what Maple is telling you with all these answers.

Our investigation of derivatives could have been speeded up substantially had we been willing to use Maple uncritically as a black box for taking limits. Here, for instance, are calculations of the derivatives of the sine, exponential, and cube root functions. It would be useful review to try to remember how we obtained these results in class. Are you sure Maple doesn't have bugs that affect these calculations?

```
> limit((sin(x+h)-sin(x))/h, h=0);
```

$$\cos(x)$$

```
> limit((exp(x+h)-exp(x))/h, h=0);
```

$$e^x$$

```
> g := x -> x^(1/3);
```

$$g := x \rightarrow x^{1/3}$$

```
> (g(x)-g(a))/(x-a);
```

$$\frac{x^{1/3} - a^{1/3}}{x - a}$$

```
> limit(%, x=a);
```

$$\frac{1}{3\, a^{2/3}}$$

For limits, as for most calculus functions, Maple has an "inert" form of the limit command that writes the limit without evaluating it. This command is called **Limit**. It's often useful in writing down formulas in Maple worksheets.

```
> Limit(x^x, x=0);
```

$$\lim_{x \to 0} x^x$$

```
> Limit(x^x, x=0) = limit(x^x, x=0);
```

$$\lim_{x \to 0} x^x = 1$$

Thinks about why the value of this limit wasn't obvious. (What's x^0? What's 0^x? What should 0^0 be, if anything?)

Derivatives

Maple has two functions for computing derivatives.

If **f** is an expression (like $x^2 \sin(x)$), then **diff(f, x)** is the derivative of **f** with respect to the variable x.

If **f** is a function (like **sin**, or like the user-defined function **g** above), then the function **D(f)** is the derivative of **f** as a function. Using **D** can save typing, and gets the derivative as a function.

```
> diff(sqrt(x+sqrt(x+sqrt(x))), x);
```

$$\frac{1}{2} \frac{1 + \frac{1}{2} \frac{1 + \frac{1}{2\sqrt{x}}}{\sqrt{x + \sqrt{x}}}}{\sqrt{x + \sqrt{x + \sqrt{x}}}}$$

```
> f := x -> x^x;
```

$$f := x \to x^x$$

```
> fprime := D(f);
```

$$fprime := x \to x^x (\ln(x) + 1)$$

```
> plot({f(x), fprime(x)}, x=0..2);
```




```
> D(sin)(x);
```

$$\cos(x)$$

```
> D(D(sin))(x);
```

$$-\sin(x)$$

```
> D(D(D(sin)))(x);
```

$$-\cos(x)$$

```
> (D@@4)(sin)(x) = D(D(D(D(sin))))(x);
```

$$\sin(x) = \sin(x)$$

These operators can also be used with undefined functions to get general differentiation rules. Right now, **f** and **g** have value, but we can remove those

```
> g(x);
```

$$x^{1/3}$$

```
> g := 'g';
```

$$g := g$$

```
> g(x);
```

$$g(x)$$

```
> f := 'f'; f(x);
```

$$f := f$$

$$f(x)$$

Here are Maple versions of the product rule, the chain rule, and an extended chain rule for $f(g(h(x)))$. Play with things like this, and make sure the notation makes sense. If you need help, try **?D** or **?diff**.

```
> D(f*g);
```

$$\mathrm{D}(f)\, g + f\,\mathrm{D}(g)$$

```
> D(f@g);
```

$$\mathrm{D}(f)@g\ \mathrm{D}(g)$$

```
> diff(f(g(h(x))), x);
```

$$\mathrm{D}(f)\,(g(h(x)))\ \mathrm{D}(g)\,(h(x))\left(\frac{\mathrm{d}}{\mathrm{d}x}\, h(x)\right)$$

Integrals

Maple does both indefinite and definite integrals with **int**. The inert form of **int** is, logically enough, **Int**. As with limits, the inert forms are useful in writing formulas.

```
> int(sin(x), x=0..Pi);
```

$$2$$

```
> int(x/(1+x^2), x);
```

$$\frac{1}{2}\ln\left(1 + x^2\right)$$

```
> int(x/(1+x^4), x);
```

$$\frac{1}{2}\arctan\left(x^2\right)$$

```
> int(x/(1+x^3), x);
```

$$\frac{1}{6}\ln\left(x^2-x+1\right)+\frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)-\frac{1}{3}\ln(1+x)$$

```
> Int(x^3*exp(x),x);
```

$$\int x^3 e^x\, dx$$

```
> Int(x^3*exp(x),x) = int(x^3*exp(x),x);
```

$$\int x^3 e^x\, dx=\left(-6+6x-3x^2+x^3\right)e^x$$

```
> Int(x^n,x) = int(x^n,x);
```

$$\int x^n\, dx=\frac{x^{n+1}}{n+1}$$

The values of inert forms can also be taken using the command **value**.

```
> Int(ln(x)^n/x, x) = value(Int(ln(x)^n/x, x));
```

$$\int \frac{\ln(x)^n}{x}\, dx=\frac{\ln(x)^{n+1}}{n+1}$$

Sums

Sums are done with **sum** and **Sum**, just as integrals are done with **int** and **Int**.

```
> sum(k, k=1..10);
```

$$55$$

```
> sum(k, k=1..n);
```

$$\frac{1}{2}(n+1)^2-\frac{1}{2}n-\frac{1}{2}$$

```
> factor(%);
```

$$\frac{1}{2}n(n+1)$$

```
> sum(k^2, k=1..n);
```

$$\frac{1}{3}(n+1)^3-\frac{1}{2}(n+1)^2+\frac{1}{6}n+\frac{1}{6}$$

```
> factor(%);
```

$$\frac{1}{6}n(n+1)(2n+1)$$

```
> Sum(k^3, k=1..n);
```

$$\sum_{k=1}^{n} k^3$$

```
> Sum(k^3, k=1..n) = factor(sum(k^3, k=1..n));
```

$$\sum_{k=1}^{n} k^3=\frac{1}{4}n^2(n+1)^2$$

The range of sums Maple understands is pretty impressive. Here's an identity that should look beautiful and surprising:

```
> Sum(1/n^2, n=1..infinity) = sum(1/n^2, n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2} = \frac{1}{6}\pi^2$$

If Maple can do all this, then Maple ought to be able to handle the nasty work we encountered with Riemann sums. Here, for instance, is Maple's treatment of a computation of

$$\int_0^x t^3 \, dt$$

using Riemann sums.

```
> f := x -> x^3;
```

$$f := x \to x^3$$

```
> RSum := Sum(f(k*x/n)*(1/n), k=1..n);
```

$$RSum := \sum_{k=1}^{n} \frac{k^3 x^3}{n^4}$$

```
> RSum = value(RSum);
```

$$\sum_{k=1}^{n} \frac{k^3 x^3}{n^4} = \frac{1}{4}\,\frac{x^3 (n+1)^4}{n^4} - \frac{1}{2}\,\frac{(n+1)^3 x^3}{n^4} + \frac{1}{4}\,\frac{x^3 (n+1)^2}{n^4}$$

```
> RSum = factor(value(RSum));
```

$$\sum_{k=1}^{n} \frac{k^3 x^3}{n^4} = \frac{1}{4}\,\frac{x^3 (n+1)^2}{n^2}$$

This works out the Riemann sum for n slices. Now we want to take the limit of this sum as $n \to \infty$.

```
> Int(t^3, t=0..x) = Limit(RSum, n=infinity);
  Int(t^3, t=0..x) = Limit(factor(value(RSum)), n=infinity);
  Int(t^3, t=0..x) = limit(factor(value(RSum)), n=infinity);
```

$$\int_0^x t^3 \, dt = \lim_{n \to \infty} \left(\sum_{k=1}^{n} \frac{k^3 x^3}{n^4} \right)$$

$$\int_0^x t^3 \, dt = \lim_{n \to \infty} \left(\frac{1}{4}\,\frac{x^3 (n+1)^2}{n^2} \right)$$

$$\int_0^x t^3 \, dt = \frac{1}{4} x^3$$

The student calculus package.

Maple also has a whole library of functions useful for the calculus student and contained in the package **Student**. We're interested in a sub-package called **Student[Calculus1]**. *You need to load the package using* **with(Student[Calculus1])** *before you can use the functions.*

```
> with(Student[Calculus1]);
```

[AntiderivativePlot, AntiderivativeTutor, ApproximateInt, ApproximateIntTutor, ArcLength, ArcLengthTutor, Asymptotes, Clear, CriticalPoints, CurveAnalysisTutor, DerivativePlot, DerivativeTutor, DiffTutor, ExtremePoints, FunctionAverage, FunctionAverageTutor, FunctionChart, FunctionPlot, GetMessage, GetNumProblems, GetProblem, Hint, InflectionPoints, IntTutor, Integrand, InversePlot, InverseTutor, LimitTutor, MeanValueTheorem, MeanValueTheoremTutor, NewtonQuotient, NewtonsMethod, NewtonsMethodTutor, PointInterpolation, RiemannSum, RollesTheorem, Roots, Rule, Show, ShowIncomplete, ShowSteps, Summand, SurfaceOfRevolution, SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem]

It's worth asking for help about most of these functions, just to see what they do. Here's a quick teaser:

The midpoint approximation to the integral of $y = \zeta(x)$ between $x = -20$ and $x = -19$ is shown below

```
> RiemannSum(Zeta(x), x=-20..-19, partition=10, method=midpoint,
  output=plot);
```



If we don't want the picture, we can write down the sum of the areas of the reactangles

```
> RiemannSum(Zeta(x), x=-20..-19, partition=10, method=midpoint);
```

$$\frac{1}{10}\zeta\left(-\frac{399}{20}\right)+\frac{1}{10}\zeta\left(-\frac{397}{20}\right)+\frac{1}{10}\zeta\left(-\frac{79}{4}\right)+\frac{1}{10}\zeta\left(-\frac{393}{20}\right)+\frac{1}{10}\zeta\left(-\frac{391}{20}\right)+\frac{1}{10}\zeta\left(-\frac{389}{20}\right)+\frac{1}{10}\zeta\left(-\frac{387}{20}\right)+\frac{1}{10}\zeta\left(-\frac{77}{4}\right)+\frac{1}{10}\zeta\left(-\frac{383}{20}\right)+\frac{1}{10}\zeta\left(-\frac{381}{20}\right)$$

and then approximate the sum numerically.

```
> % = evalf(value(%));
```

$$\frac{1}{10}\zeta\left(-\frac{399}{20}\right)+\frac{1}{10}\zeta\left(-\frac{397}{20}\right)+\frac{1}{10}\zeta\left(-\frac{79}{4}\right)+\frac{1}{10}\zeta\left(-\frac{393}{20}\right)+\frac{1}{10}\zeta\left(-\frac{391}{20}\right)+\frac{1}{10}\zeta\left(-\frac{389}{20}\right)+\frac{1}{10}\zeta\left(-\frac{387}{20}\right)+\frac{1}{10}\zeta\left(-\frac{77}{4}\right)+\frac{1}{10}\zeta\left(-\frac{383}{20}\right)+\frac{1}{10}\zeta\left(-\frac{381}{20}\right)=26.63867649$$

By comparison, a more exact value of the integral is

```
> Int(Zeta(x), x=-20..-19) = evalf(int(Zeta(x), x=-20..-19));
```

$$\int_{-20}^{-19}\zeta(x)\,\mathrm{d}x=26.57108226$$

The error in the midpoint sum is therefore only

```
> RiemannSum(Zeta(x), x=-20..-19, partition=10, method=midpoint) -
  Int(Zeta(x), x=-20..-19) =
  evalf(value(RiemannSum(Zeta(x), x=-20..-19, partition=10,
  method=midpoint))-
  evalf(int(Zeta(x), x=-20..-19));
```

$$\frac{1}{10}\zeta\left(-\frac{399}{20}\right)+\frac{1}{10}\zeta\left(-\frac{397}{20}\right)+\frac{1}{10}\zeta\left(-\frac{79}{4}\right)+\frac{1}{10}\zeta\left(-\frac{393}{20}\right)+\frac{1}{10}\zeta\left(-\frac{391}{20}\right)+\frac{1}{10}\zeta\left(-\frac{389}{20}\right)+\frac{1}{10}\zeta\left(-\frac{387}{20}\right)+\frac{1}{10}\zeta\left(-\frac{77}{4}\right)+\frac{1}{10}\zeta\left(-\frac{383}{20}\right)+\frac{1}{10}\zeta\left(-\frac{381}{20}\right)-\left(\int_{-20}^{-19}\zeta(x)\,\mathrm{d}x\right)=0.06759423$$

Integration Tutor

One of the neatest uses of the **Student[Calculus1]** package is that it lets one explore possible choices of u when doing substitution and choices of u and v' when doing integration by parts. This is easier to see on your own than it is to explain here, but here are the commands you would use to open the integration tutor for the integrals

$$\int \frac{x^2}{1+x^6}\,\mathrm{d}x$$

and

$$\int x^3 \cdot e^{(x^2)} \mathrm{d}x.$$

```
> IntTutor(x^2/(1+x^6),x);
> IntTutor(x^3*exp(x^2),x);
```

You might enjoy exploring, for instance, what happens with the first integral if you let $u = x^2$ and then if you let $u = x^3$. The only tricky bit might be that *Maple* calls substitution "change of variable," so that in the IntTutor you do it by hitting the **Change** button. You might also enjoy exploring Parts as a technique for the second integral, letting u be x^3, then trying $u = \exp(x^2)$, then $u = x$, then $u = x^2$. The **Undo** button will be useful in this play.

There is lots more functionality in *Maple* that is relevant to calculus, and there are lots more fun and useful functions in the **Student[Calculus1]** package. A bit of time spent playing with *Maple's* tutorials and messing about with the online help can be very rewarding.

Sample Problems to play with.

If you want to mess about and to get comfortable, here are some problems you migt have fun with.

1. Use Maple to explore the sums

$$\sum_{n=1}^{\infty} \frac{1}{n^s}$$

for various values of s. What do you observe?

2. As efficiently as possible, compute the 10th derivative of $\sec(x)$.

3. What happens when you use the substitution $u = \arcsin(x)$ on the integral

$$\int \sqrt{1 - x^2}\, \mathrm{d}x?$$

What if you use $u = \arctan(x)$ on the integral

$$\int \sqrt{1 + x^2}\, \mathrm{d}x?$$

Can you understand what's going on by doing these integrals by hand? Can you find other integral where these substitutions work? Be creative!

4. Use Maple to explore sums like

$$\sum_{k=1}^{n} k^s$$

for various values of s, and see if you observe anything. Also, use Maple to look at sums like

$$\sum_{k=1}^{n} k\,(k+1),\quad \sum_{k=1}^{n} k\,(k+1)\,(k+2),\quad \sum_{k=1}^{n} k\,(k+1)\,(k+2)\,(k+3), \ldots$$

See if you notice anything here. It might help if you factor the expressions you compute.

5. Let f(x) be a parabola

```
> f := x -> a*x^2 + b*x + c;
```

$$f := x \to a\,x^2 + b\,x + c$$

What must *a*, *b*, and *c* be in order for *f* to have the same value, the same first derivative, and the same second derivative as $\sin(x)$ at $x = 0$? (COmpute the derivatives of both functions at 0, set them equal to one another, and solve the resulting system of equations.) Plot sin(*x*) together with this parabola.

Now take a generic cubic, $f(x) = a\,x^3 + b\,x^2 + c\,x + d$, and try to make its value and its first, second, and third derivatives equal to the value and the first 3 derivatives of sin(*x*) at 0. Again, plot this cubic together with sin(*x*).

Keep doing this with polynomials of increasing degree. You should be getting curves that approximate the sine function better and better.

6. Ask Maple about the sum

$$\sum_{n=0}^{\infty} \frac{(-1)^n x^{2n+1}}{(2n+1)!}.$$

What has this got to do with Problem 5?